

Description

To demonstrate a structural model with a measurement component, we use data from [Wheaton et al. \(1977\)](#):

```
. use https://www.stata-press.com/data/r19/sem_sm2  
(Structural model with measurement component)  
. ssd describe  
Summary statistics data from  
https://www.stata-press.com/data/r19/sem_sm2.dta  
Observations: 932 Structural model with measurem..  
Variables: 13 25 May 2024 11:45  
(_dta has notes)  


---



| Variable name | Variable label                       |
|---------------|--------------------------------------|
| educ66        | Education, 1966                      |
| occstat66     | Occupational status, 1966            |
| anomia66      | Anomia, 1966                         |
| pwless66      | Powerlessness, 1966                  |
| socdist66     | Latin American social distance, 1966 |
| occstat67     | Occupational status, 1967            |
| anomia67      | Anomia, 1967                         |
| pwless67      | Powerlessness, 1967                  |
| socdist67     | Latin American social distance, 1967 |
| occstat71     | Occupational status, 1971            |
| anomia71      | Anomia, 1971                         |
| pwless71      | Powerlessness, 1971                  |
| socdist71     | Latin American social distance, 1971 |



---



```
. notes
_dta:
1. Summary statistics data from Wheaton, B., Muthen B., Alwin, D., &
Summers, G., 1977, "Assessing reliability and stability in panel models",
in D. R. Heise (Ed.), _Sociological Methodology 1977_ (pp. 84-136), San
Francisco: Jossey-Bass, Inc.
2. Four indicators each measured in 1966, 1967, and 1981, plus another
indicator (educ66) measured only in 1966.
3. Intended use: Create structural model relating Alienation in 1971,
Alienation in 1967, and SES in 1966.
```


```

See [Structural models 9: Unobserved inputs, outputs, or both](#) in [\[SEM\] Intro 5](#) for background.


Remarks and examples

Remarks are presented under the following headings:

- [Fitting the model](#)
- [Fitting the model with the Builder](#)
- [Evaluating omitted paths with estat mindices](#)
- [Refitting the model](#)

Fitting the model

Simplified versions of the model fit by the authors of the referenced paper appear in many SEM software manuals. A simplified model is



```

. sem
>     (anomia67 pwless67 <- Alien67)      ///
>     (anomia71 pwless71 <- Alien71)      ///
>     (Alien67 <- SES)                      ///
>     (Alien71 <- Alien67 SES)              ///
>     (    SES -> educ66 occstat66)        // measurement piece
                                         // structural piece
                                         // structural piece
                                         // measurement piece
  
```

Endogenous variables

Measurement: anomia67 pwless67 anomia71 pwless71 educ66 occstat66
 Latent: Alien67 Alien71

Exogenous variables

Latent: SES

Fitting target model:

Iteration 0: Log likelihood = -15249.988
 Iteration 1: Log likelihood = -15246.584
 Iteration 2: Log likelihood = -15246.469
 Iteration 3: Log likelihood = -15246.469

Example 9 — Structural model with measurement component 3

Structural equation model
Estimation method: ml

Number of obs = 932

Log likelihood = -15246.469

- (1) [anomia67]Alien67 = 1
- (2) [anomia71]Alien71 = 1
- (3) [educ66]SES = 1

	OIM					
	Coefficient	std. err.	z	P> z	[95% conf. interval]	
Structural Alien67 SES	-.6140404	.0562407	-10.92	0.000	-.7242701	-.5038107
Alien71 Alien67 SES	.7046342 -.1744153	.0533512 .0542489	13.21 -3.22	0.000 0.001	.6000678 -.2807413	.8092007 -.0680894
Measurement anomia67 Alien67 _cons		1 (constrained)				
	13.61	.1126205	120.85	0.000	13.38927	13.83073
pwless67 Alien67 _cons	.8884887 14.67	.0431565 .1001798	20.59 146.44	0.000 0.000	.8039034 14.47365	.9730739 14.86635
anomia71 Alien71 _cons		1 (constrained)				
	14.13	.1158943	121.92	0.000	13.90285	14.35715
pwless71 Alien71 _cons	.8486022 14.9	.0415205 .1034537	20.44 144.03	0.000 0.000	.7672235 14.69723	.9299808 15.10277
educ66 SES _cons		1 (constrained)				
	10.9	.1014894	107.40	0.000	10.70108	11.09892
occstat66 SES _cons	5.331259 37.49	.4307503 .6947112	12.38 53.96	0.000 0.000	4.487004 36.12839	6.175514 38.85161
var(e.ano~67) var(e.pwl~67) var(e.ano~71) var(e.pwl~71) var(e.educ66) var(e.occ~66) var(e.Ali~67) var(e.Ali~71) var(SES)	4.009921 3.187468 3.695593 3.621531 2.943819 260.63 5.301416 3.737286 6.65587	.3582978 .283374 .3911512 .3037908 .5002527 18.24572 .483144 .3881546 .6409484		3.365724 2.677762 3.003245 3.072483 2.109908 227.2139 4.434225 3.048951 5.511067		4.777416 3.794197 4.54755 4.268693 4.107319 298.9605 6.338201 4.581019 8.038482

LR test of model vs. saturated: chi2(6) = 71.62

Prob > chi2 = 0.0000

Notes:

1. Measurement component: In both 1967 and 1971, anomia and powerlessness are used to measure endogenous latent variables representing alienation for the same two years. Education and occupational status are used to measure the exogenous latent variable SES.

2. Structural component: SES->Alien67 and SES->Alien71, and Alien67->Alien71.
3. The model versus saturated χ^2 test indicates that the model is a poor fit.

Fitting the model with the Builder

Use the diagram above for reference.

1. Open the dataset.

In the Command window, type

```
. use https://www.stata-press.com/data/r19/sem_sm2
```

2. Open a new Builder diagram.

Select menu item **Statistics > SEM (structural equation modeling) > Model building and estimation.**

3. Create the measurement component for alienation in 1967.

Select the Add measurement component tool, , and then click in the diagram about halfway down from the top and one-third of the way in from the left.

In the resulting dialog box,

- a. change the *Latent variable name* to Alien67;
- b. select anomia67 and pwless67 by using the *Measurement variables* control;
- c. select Up in the *Measurement direction* control;
- d. click on **OK**.

If you wish, move the component by clicking on any variable and dragging it.

4. Create the measurement component for alienation in 1971.

Repeat the process from item 3, but place the measurement component about halfway down from the top and two-thirds of the way in from the left. Label the latent variable Alien71, and select measurement variables anomia71 and pwless71. Again drag to reposition if desired.

5. Create the measurement component for socioeconomic status.

Select the Add measurement component tool, , and then click in the diagram about one-third of the way up from the bottom and one-fourth of the way in from the left.

In the resulting dialog box,

- a. change the *Latent variable name* to SES;
- b. select educ66 and occstat66 by using the *Measurement variables* control;
- c. select Down in the *Measurement direction* control;
- d. click on **OK**.

6. Create a path from Alien67 to Alien71.

- a. Select the Add path tool, .
- b. Click in the right half of the Alien67 oval (it will highlight when you hover over it), and drag a path to the left half of the Alien71 oval (it will highlight when you can release to connect the path).

7. Create a path from SES to Alien67.

Continuing with the  tool, click in the upper portion of the SES oval, and drag a path to the lower-left quadrant of the Alien67 oval.

8. Create a path from SES to Alien71.

Continuing with the  tool, click in the upper-right quadrant of the SES oval, and drag a path to the lower-left quadrant of the Alien71 oval.

9. Clean up the direction of the errors.

The error on Alien67 may have been created under the path from Alien67 to Alien71. If so, choose the Select tool, , and then click in the Alien67 oval. Click on one of the **Error rotation** buttons, , in the Contextual Toolbar until the error is where you want it.

10. Estimate.

Click on the **Estimate** button, , in the Standard Toolbar, and then click on **OK** in the resulting **SEM estimation options** dialog box.

Tips: We took some care above to click in specific quadrants of the ovals when creating paths. This is because, by default, the Builder places the ends of paths on the boundary of the ovals and rectangles closest to where we click. When you create paths, it is often convenient to change the automation settings (**Settings > Automation...**) so that *Connection attachment points* is *Attach based on position of variables* rather than *Attach nearest to drop point*. With the *Connection attachment points* setting, the paths connect nicely regardless of where we click or drop in the variables' ovals and rectangles.

Note that when you create covariances with the Covariances tool, , *Attach nearest to drop point* is sometimes preferred.

You can open a completed diagram in the Builder by typing

```
. webgetsem sem_sm2
```

Evaluating omitted paths with estat mindices

That the model is a poor fit leads us to look at the modification indices:

```
. estat mindices
Modification indices
```

	MI	df	P>MI	Standard EPC	
Measurement anomia67					
anomia71	51.977	1	0.00	.3906425	.4019984
pwless71	32.517	1	0.00	-.2969297	-.2727609
educ66	5.627	1	0.02	.0935048	.0842631
pwless67					
anomia71	41.618	1	0.00	-.3106995	-.3594367
pwless71	23.622	1	0.00	.2249714	.2323233
educ66	6.441	1	0.01	-.0889042	-.0900664
anomia71					
anomia67	58.768	1	0.00	.429437	.4173061
pwless67	38.142	1	0.00	-.3873066	-.3347904
pwless71					
anomia67	46.188	1	0.00	-.3308484	-.3601641
pwless67	27.760	1	0.00	.2871709	.2780833
educ66					
anomia67	4.415	1	0.04	.1055965	.1171781
pwless67	6.816	1	0.01	-.1469371	-.1450411
cov(e.anomia67,e.anomia71)	63.786	1	0.00	1.951578	.5069627
cov(e.anomia67,e.pwless71)	49.892	1	0.00	-1.506704	-.3953794
cov(e.anomia67,e.educ66)	6.063	1	0.01	.5527612	.1608845
cov(e.pwless67,e.anomia71)	49.876	1	0.00	-1.534199	-.4470094
cov(e.pwless67,e.pwless71)	37.357	1	0.00	1.159123	.341162
cov(e.pwless67,e.educ66)	7.752	1	0.01	-.5557802	-.1814365

EPC is expected parameter change.

Notes:

1. There are lots of statistically significant paths we could add to the model.
2. Some of those statistically significant paths also make theoretical sense.
3. Two in particular that make theoretical sense are the covariances between e.anomia67 and e.anomia71 and between e.pwless67 and e.pwless71.

Refitting the model

Let's refit the model and include those two previously excluded covariances:

```
. sem
>     (anomia67 pwless67 <- Alien67)           ///
>     (anomia71 pwless71 <- Alien71)           ///
>     (Alien67 <- SES)                         ///
>     (Alien71 <- Alien67 SES)                 ///
>     (      SES -> educ66 occstat66)          ///
>             , cov(e.anomia67*e.anomia71)    ///
>             cov(e.pwless67*e.pwless71)

Endogenous variables
Measurement: anomia67 pwless67 anomia71 pwless71 educ66 occstat66
Latent:      Alien67 Alien71

Exogenous variables
Latent: SES

Fitting target model:
Iteration 0: Log likelihood = -15249.988
Iteration 1: Log likelihood = -15217.95
Iteration 2: Log likelihood = -15213.126
Iteration 3: Log likelihood = -15213.046
Iteration 4: Log likelihood = -15213.046
```

Example 9 — Structural model with measurement component 8

Structural equation model
Estimation method: ml

Number of obs = 932

Log likelihood = -15213.046

- (1) [anomia67]Alien67 = 1
- (2) [anomia71]Alien71 = 1
- (3) [educ66]SES = 1

	OIM					
	Coefficient	std. err.	z	P> z	[95% conf. interval]	
Structural Alien67 SES	-.5752228	.057961	-9.92	0.000	-.6888244	-.4616213
Alien71 Alien67 SES	.606954 -.2270301	.0512305 .0530773	11.85 -4.28	0.000 0.000	.5065439 -.3310596	.707364 -.1230006
Measurement anomia67 Alien67 _cons		1 (constrained)				
	13.61	.1126143	120.85	0.000	13.38928	13.83072
pwless67 Alien67 _cons	.9785952 14.67	.0619825 .1001814	15.79 146.43	0.000 0.000	.8571117 14.47365	1.100079 14.86635
anomia71 Alien71 _cons		1 (constrained)				
	14.13	.1159036	121.91	0.000	13.90283	14.35717
pwless71 Alien71 _cons	.9217508 14.9	.0597225 .1034517	15.43 144.03	0.000 0.000	.8046968 14.69724	1.038805 15.10276
educ66 SES _cons		1 (constrained)				
	10.9	.1014894	107.40	0.000	10.70108	11.09892
occstat66 SES _cons	5.22132 37.49	.425595 .6947112	12.27 53.96	0.000 0.000	4.387169 36.12839	6.055471 38.85161
var(e.ano~67) var(e.pwl~67) var(e.ano~71) var(e.pwl~71) var(e.educ66) var(e.occ~66) var(e.Ali~67) var(e.Ali~71) var(SES)	4.728874 2.563413 4.396081 3.072085 2.803674 264.5311 4.842059 4.084249 6.796014	.456299 .4060733 .5171156 .4360333 .5115854 18.22483 .4622537 .4038995 .6524866			3.914024 1.879225 3.490904 2.326049 1.960691 231.1177 4.015771 3.364613 5.630283	5.713365 3.4967 5.535966 4.057398 4.009091 302.7751 5.838364 4.957802 8.203105
cov(e.ano~67, e.anomia71) cov(e.pwl~67, e.pwless71)	1.622024 .3399961	.3154267 .2627541	5.14 1.29	0.000 0.196	1.003799 -.1749925	2.240249 .8549847

LR test of model vs. saturated: chi2(4) = 4.78

Prob > chi2 = 0.3111

Notes:

1. We find the covariance between e.anomia67 and e.anomia71 to be significant ($Z = 5.14$).
2. We find the covariance between e.pwless67 and e.pwless71 to be insignificant at the 5% level ($Z = 1.29$).
3. The model versus saturated χ^2 test indicates that the model is a good fit.

References

- Acock, A. C. 2013. *Discovering Structural Equation Modeling Using Stata*. Rev. ed. College Station, TX: Stata Press.
- Wheaton, B., B. Muthén, D. F. Alwin, and G. F. Summers. 1977. “Assessing reliability and stability in panel models”. In *Sociological Methodology* 1977, edited by D. R. Heise, 84–136. San Francisco: Jossey-Bass.

Also see

- [SEM] **Example 32g** — Full structural equation model (generalized response)
- [SEM] **estat mindices** — Modification indices
- [SEM] **test** — Wald test of linear hypotheses

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.

For suggested citations, see the FAQ on [citing Stata documentation](#).

